

## Q&R : Questions / Requêtes

#by Aymeric Castellanet

### 1. Quel est le nombre de stations par nombre de correspondances ?

Avec une première requête, on récupère pour chaque station, son nombre de correspondances, ordonné par nombre de correspondances décroissant :

```
MATCH (s:Station)
RETURN s.nom AS nom_station, COUNT(s.nom_maj) AS nb_correspondances
ORDER BY nb_correspondances DESC
```

"nom_station"	"nb_correspondances"
"Châtelet"	5
"République"	5
"Montparnasse-Bienvenüe"	4
"Nation"	4
"Saint-Lazare"	4
"Bastille"	3
"Charles de Gaulle-Étoile"	3
"Concorde"	3
"Gare de l'Est"	3
"Jaurès"	3
"La Motte-Picquet-Grenelle"	3

On réutilise ce résultat dans une requête imbriquée à l'aide du mot-clé WITH, puis on retourne le décompte du nombre de stations pour chaque nombre de correspondances :

```
MATCH (s:Station)
WITH s.nom AS nom_station, COUNT(s.nom_maj) AS nb_correspondances
RETURN nb_correspondances, COUNT(nom_station) AS nb_stations
ORDER BY nb_correspondances DESC
```


"nb_correspondances"	"nb_stations"
5	2
4	3
3	11
2	42
1	244

## 2. Quel est le nombre de stations à moins de deux kilomètres de la station LADEFENSE ?

Pas beaucoup vu que La Grande Arche n'est que sur la ligne 1 et que c'est son terminus (puisque dans notre modèle, on ne prend pas en compte les RER ni les trains de banlieue) ...

... Maintenant en demandant à Cypher de manières plus précises de me retourner les stations :

```
MATCH (s0:Station {nom_maj: "LADEFENSE"})
MATCH (s:Station)
WHERE distance(s0.coordonnees, s.coordonnees) < 2000
RETURN s0, s
```



Visuellement, on en voit deux : Esplanade et Pont de Neuilly.

Et en faisant le calcul avec un COUNT() -1 (-1 pour retirer la station LADEFENSE elle-même du décompte) :

```
MATCH (s0:Station {nom_maj: "LADEFENSE"})
MATCH (s:Station)
WHERE distance(s0.coordonnees, s.coordonnees) < 2000
RETURN s0.nom_maj, COUNT(s.nom_maj) - 1 AS stations_proches
```

"s0.nom_maj"	"stations_proches"
"LADEFENSE"	2

### 3. Combien de temps faut-il pour aller en métro de LADEFENSE à CHATEAUDEVINCENNES ?

Cette première requête nous permet d'obtenir pour les deux stations la ligne de métro associée ainsi que la distance en mètres entre ces deux stations :

```
MATCH (s1:Station {nom_maj: "LADEFENSE"})
MATCH (s2:Station {nom_maj: "CHATEAUDEVINCENNES"})
RETURN [s1.nom_maj, s1.ligne, s1.coordonnees],
       [s2.nom_maj, s2.ligne, s2.coordonnees],
       distance(s1.coordonnees, s2.coordonnees) AS distance
```

"[s1.nom_maj, s1.ligne, s1.coordonnees]"	"[s2.nom_maj, s2.ligne, s2.coordonnees]"	"distance"
["LADEFENSE", "1", point({srid:7203, x:644056.2663, y:6866067.7406})]	["CHATEAUDEVINCENNES", "1", point({srid:7203, x:658933.7619, y:6860635.4943})]	15838.218813739164

On observe que les deux stations sont sur la même ligne de métro (la ligne 1), et donc qu'il n'y a pas de changements à faire qui rajouterai du temps.

Le temps de transport en métro entre ces deux stations peut donc se calculer en divisant la distance par la vitesse moyenne du métro (40 km/h), soit le calcul suivant :  $15838/40000*60 = 23,8$  min

Ou alors avec la requête imbriquée suivante :

```
MATCH (s1:Station {nom_maj: "LADEFENSE"})
MATCH (s2:Station {nom_maj: "CHATEAUDEVINCENNES"})
WITH s1.nom_maj AS station_départ, s2.nom_maj AS station_arrivée,
     distance(s1.coordonnees, s2.coordonnees) AS distance
RETURN station_départ, station_arrivée, distance,
       distance/40000*60 AS temps_métro
```

"station_départ"	"station_arrivée"	"distance"	"temps_métro"
"LADEFENSE"	"CHATEAUDEVINCENNES"	15838.218813739164	23.75732822060875

#### 4. Combien de temps faut-il pour aller à pied de LADEFENSE à CHATEAUDEVINCENNES ?

On repart de la requête précédente en divisant la distance par la vitesse moyenne de marche à pied d'une personne (4 km/h), soit le calcul suivant :  $15838/4000*60 = 238 \text{ min} = 3\text{h}58\text{min}$

Ou alors avec la requête imbriquée suivante :

```
MATCH (s1:Station {nom_maj: "LADEFENSE"})
MATCH (s2:Station {nom_maj: "CHATEAUDEVINCENNES"})
WITH s1.nom_maj AS station_départ, s2.nom_maj AS station_arrivée,
     distance(s1.coordonnees, s2.coordonnees) AS distance
RETURN station_départ, station_arrivée, distance,
       distance/4000*60 AS temps_marche_minutes
```

"station_départ"	"station_arrivée"	"distance"	"temps_marche_minutes"
"LADEFENSE"	"CHATEAUDEVINCENNES"	15838.218813739164	237.57328220608747

#### 5. Faut-il faire un changement à STLAZARE pour aller de MONTPARNASSEBIENVENUE à GABRIELPERI ?

On recherche si MONTPARNASSEBIENVENUE et GABRIELPERI sont sur une même ligne de métro.


```
MATCH (s1:Station {nom_maj: "MONTPARNASSEBIENVENUE"})
MATCH (s2:Station {nom_maj: "GABRIELPERI"})
WHERE s1.ligne = s2.ligne
RETURN s1.nom_maj, s2.nom_maj, s1.ligne AS ligne_commune
```

"s1.nom_maj"	"s2.nom_maj"	"ligne_commune"
"MONTPARNASSEBIENVENUE"	"GABRIELPERI"	"13"

Ces deux stations étant sur une même ligne (la ligne 13), il n'y a pas de changement à faire.

Si on se pose la question de : est-ce le trajet par métro pour aller de MONTPARNASSEBIENVENUE à GABRIELPERI, en utilisant la ligne commune ces deux stations, passe par STLAZARE, on peut faire la requête suivante :

```
MATCH (s1:Station {nom_maj: "MONTPARNASSEBIENVENUE"})
MATCH (s2:Station {nom_maj: "STLAZARE"})
MATCH (s3:Station {nom_maj: "GABRIELPERI"})
WHERE s1.ligne = s3.ligne
MATCH p = (s1)-[:LIAISON_TRAIN*..10]->(s2)-[:LIAISON_TRAIN*..10]->(s3)
RETURN p
```




On observe visuellement que la station Saint Lazare est effectivement entre Montparnasse et Gabriel Péri.

On aurait aussi pu répondre à cette question à l'aide de l'algorithme de Dijkstra de Neo4j qui nous permet de rechercher le chemin le plus court entre MONTPARNASSEBIENVENUE et GABRIELPERI :

```
MATCH (start:Station {nom_maj: "MONTPARNASSEBIENVENUE"})
MATCH (end:Station {nom_maj: "GABRIELPERI"})
CALL gds.alpha.shortestPath.stream({
  nodeQuery: "MATCH (n) RETURN id(n) as id",
  relationshipQuery: "MATCH (n1)-[r]-(n2) RETURN id(r) AS id, id(n1) AS
source, id(n2) AS target",
  startNode: start,
  endNode: end})
YIELD nodeId
RETURN gds.util.asNode(nodeId)
```

neo4j\$ MATCH (start:Station {nom\_maj: "MONTPARNASSEBIENVENUE"}) MATCH (e



6. Combien de stations se trouvent dans un rayon de 10 stations par train autour de STLAZARE ?

```
MATCH (s0:Station {nom_maj: "STLAZARE"})
MATCH (s0)-[:LIAISON_TRAIN*..10]->(s:Station)
RETURN COUNT(DISTINCT s.nom_maj) - 1 AS nb_stations
```

neo4j\$ MATCH (s0:St

nb_stations
71

7. Combien de stations se trouvent dans un rayon de 20 minutes par train autour de STLAZARE ?

20 minutes par train à une vitesse de 40 km/h correspond à distance de :

$20 \times 40 / 60 = 13,33 \text{ km} = 13\,330 \text{ m}$

```
MATCH (s0:Station {nom_maj: "STLAZARE"})
MATCH (s0)-[:LIAISON_TRAIN*1..]->(s:Station)
  WHERE distance(s0.coordonnees, s.coordonnees) < 13330
RETURN COUNT(DISTINCT s.nom_maj) - 1 AS nb_stations
```

neo4j\$ MATCH (s0:Sta

nb_stations
89